6.4 One Householder Transformation

A. Purpose
These subroutines compute the parameters defining a Householder orthogonal transformation which zeroes specified components in a given vector, or in a column or row of a matrix. Also they optionally apply either a newly defined transformation or a previously defined Householder transformation to a vector or a set of vectors, which typically would be columns or rows of a matrix.

Two versions are provided. For the usual case in which both the pivot vector and the vectors to which the transformation is to be applied are column vectors, we recommend the use of DHTCC or SHTCC. If either the pivot vector or the vectors to which the transformations are to be applied, or both, are row vectors, one must use DHTGEN or SHTGEN.

These subroutines are used by other library subroutines, and can be used as modules in implementing other linear algebra algorithms.

B. Usage
B.1 Usage of SHTGEN and DHTGEN
We use the term pivot vector to mean the vector that plays a special role in determining or defining the Householder transformation. When MODE = 1 a transformation will be determined that is appropriate to zero components L1 through M of the pivot vector. These components are then replaced by values that partially define the appropriate Householder transformation. Component LPivot of the pivot vector, and UPARAM are also assigned values to complete the definition of the transformation. After defining the transformation it will optionally be applied to specified vectors.

When MODE = 2 the transformation defined in a previous call with MODE = 1 will be applied to specified vectors.

B.1.a Program Prototype, Single Precision
INTEGER MODE, LPivot, L1, M, LDU, LDC, NCV
LOGICAL COLU, COLC
REAL U(LDU,*), UPARAM, C(LDC,*)
Assign values to all integer and logical arguments and to the relevant real arguments.

CALL SHTGEN(MODE, LPivot, L1, M, U, LDU, COLU, UPARAM, C, LDC, NCV, COLC)

Computed results will be returned in U(,), UPARAM, and/or C(,), depending on the initial settings of the integer arguments.

B.1.b Argument Definitions

MODE [in] Set by the user to the value 1 or 2. If MODE = 1, the subroutine computes parameters defining a Householder transformation as described in Section D. If MODE = 2, it is assumed that Householder transformation parameters have already been defined by a previous call with MODE = 1. In either case, if NCV > 0, the subroutine will apply the current transformation to the set of NCV M-vectors stored in the array C(,).

LPivot, L1, M [in] These integers specify that the vectors to be referenced or operated upon are M-dimensional, and the components to be referenced or operated upon in each vector are the one indexed by LPivot and the set indexed from L1 through M. To define a nontrivial transformation these integers must satisfy

1 ≤ LPivot < L1 ≤ M

If these inequalities are not all satisfied, the subroutine returns without doing any computation. This is not regarded as an error. It has the implicit effect of applying an identity transformation.

U(,) [inout] The array U(,) contains the pivot vector as either a column or a row. Examples:

If the pivot vector is column J of an array declared as A(IDC,IDJC), the arguments corresponding to “U, LDU, COLU” should be written as “A(1, J), IDUC, .true.”.

If the pivot vector is row I of an array declared as A(IDC,IDJC), the arguments corresponding to “U, LDU, COLU” should be written as “A(I,1), IDUC, .false.”.

LDU [in] Leading dimensioning parameter for U(,).

COLU [in] Must be set to .true. if the pivot vector is a column of U(,) and to .false. if the pivot vector is a row of U(,).

UPARAM [inout] Holds a value completing the definition of a Householder transformation. See description in Section D. This value will be computed by the subroutine when MODE = 1 and must be present on entry when MODE = 2.
C(,)  [inout] If NCV ≤ 0, no reference will be made to C(,). If NCV > 0, the array C(,) is regarded as containing a set of NCV M-vectors. These are regarded as column vectors if COLC = .true. and row vectors if COLC = .false. On entry C(,) contains vectors to be transformed, and on return C(,) contains the vectors resulting from application of the current Householder transformation. Examples:

To apply a transformation to columns J + 1 through M1 of an array declared as A(IDIM,JDIM), the arguments “C, LDC, NCV, COLC” should be written as “A(K+1, 1), IDIM, M1−J, .true.”.

To apply a transformation to rows K + 1 through N of an array declared as A(IDIM,JDIM), the arguments “C, LDC, NCV, COLC” should be written as “A(K+1, 1), IDIM, N−K, .false.”.

LDC  [in] Leading dimensioning parameter of C(,).

NCV  [in] Number of vectors in C(,) to be transformed. If NCV ≤ 0, the array C(,) will not be referenced.

COLC  [in] Must be set to .true. if the vectors to be transformed are column vectors of C(,), and to .false. if the vectors to be transformed are row vectors of C(,).

B.2 Usage of SHTCC and DHTCC

Subroutine DHTCC is a modification of DHTGEN specialized for the case of COLU = .true. and COLC = .true.

B.2.a Program Prototype, Single Precision

INTEGER  MODE, LPIVOT, L1, M, LDC, NCV
REAL  U(*), UPARAM, C(LDC,*)

Assign values to all integer and logical arguments and to the relevant real arguments.

CALL SHTCC(MODE, LPIVOT, L1, M, U, UPARAM, C, LDC, NCV)

Computed results will be returned in U(,), UPARAM, and/or C(,), depending on the initial settings of the integer arguments.

B.2.b Argument Definitions

The arguments have the same meanings as the arguments of the same names in the call to SHTGEN. Note that arguments LDU, COLU, and COLC are not used, and U is a one-dimensional array. This subroutine functions as though COLU = .true. and COLC = .true.

B.3 Modifications for Double Precision

For double precision usage change the REAL statement to DOUBLE PRECISION and change the subroutine names SHTGEN and SHTCC to DHTGEN and DHTCC respectively.

C. Examples and Remarks

The program DRSHTCC illustrates the use of SHTCC. The array D(,) is initialized with a 3 × 2 matrix, A, in its first 2 columns, and a 3 × 3 identity matrix in columns 3 through 5. Subroutine SHTCC is used to compute the QR factorization of A, i.e., the program determines an orthogonal matrix, Q, and a triangular matrix, R, such that $A = QR$. Results are shown in ODSHTCC.

For additional examples of the use of SHTCC and SHTGEN see the source code for the library subroutines SHFTI or SSVDRS.

D. Functional Description

Let $u$ be a nonzero n-vector and define $\beta = -||u||^2/2$. The matrix $Q = I + \beta^{-1}uu^t$ is symmetric and orthogonal, and is called a Householder transformation matrix.

A product of the form $w = Qv$ can be computed by the steps:

$$\gamma = (u^tv)/\beta$$

$$w = v + \gamma u$$

Typically one desires a $u$ such that for some particular nonzero vector, $v$, the vector $w = Qv$ will consist of zeros except for the first component. Given $v$, the appropriate $u$ and $\beta$ to accomplish this can be defined mathematically as follows:

$$\sigma = 1$$  if $v_1 > 0$ and $-1$ otherwise

$$u_1 = v_1 + \sigma||v||$$

$$u_i = v_i, \quad i = 2, ..., n$$

$$\beta = -\sigma u_1 ||v||$$  (Note that $\beta \leq 0$.)

With this definition of $\sigma$, $u$, and $\beta$, and thus of $Q$, the vector $w = Qv$ will have components:

$$w_1 = -\sigma||v||$$

$$w_i = 0, \quad i = 2, ..., n.$$

In applications it is frequently convenient to regard the vectors $v$, $u$, and $w$ of the above formulas as embedded in higher dimensional vectors with other components which play no role in the current transformation. Specifically we regard the host vector for $v$ (and similarly for $u$ and $w$) as being of dimension M, with $v_1$ at position LPIVOT and $v_2, ..., v_n$ in positions L1 through M. Thus we are
identifying $n$ with the value $M - L1 + 2$. We assume $1 \leq \text{LPIVOT} < L1 \leq M$.

On input with $\text{MODE} = 1$ the array $U(,)$ contains $v$ embedded as specified by the integers $\text{LPIVOT}$, $L1$, and $M$, and with the Fortran 77 storage mapping as specified by the arguments $\text{LDU}$ and $\text{COLU}$. The subroutine computes $w_1$ and $u$. It stores the $u_1$ in $\text{UPARAM}$, $w_1$ in place of $v_1$, and $u_2, ..., u_M$ in place of $v_2, ..., v_M$.

When applying transformations the quantity $\beta$ is computed as $\beta = u_1w_1$, using the value of $u_1$ stored in $\text{UPARAM}$ and $w_1$ stored in $U(,)$. This is essentially Algorithm H12 of [1] with a change in the way the user specifies the column/row options.

This is essentially Algorithm H12 of [1] with a change in the way the user specifies the column/row options.

References

E. Error Procedures and Restrictions
To define and/or apply a nontrivial transformation one must have

$$1 \leq \text{LPIVOT} < L1 \leq M$$

If these conditions are not all satisfied the subroutine returns immediately, doing no computation. This is not regarded as an error condition as it may occur intentionally as an end condition in a loop. It will have the effect of computing and/or applying an identity transformation.

If, on entry with $\text{MODE} = 1$, the components of the pivot vector indexed by $\text{LPIVOT}$, and $L1$ through $M$, are all zero, the subroutine will effectively define an identity transformation. It will set $\text{UPARAM} = 0$. This is not regarded as an error condition.

F. Supporting Information
The source language is ANSI Fortran 77.

Entry Required Files
DHTCC DHTCC, DNRM2
DHTGEN DAXPY, DDOT, DHTGEN, DNRM2
SHTCC SHTCC, SNRM2
SHTGEN SAXPY, SDOT, SHTGEN, SNRM2

DRSHTCC

program DRSHTCC

c> 1994-10-19 DRSHTCC Krogh Changes to use M77CON

c> 1987-12-09 DRSHTCC Lawson Initial Code.

c Demo driver for SHTCC, Householder transformations.

---S replaces "?": DR?HTCC, ?HTCC, ?MATP

integer IDIM, JDIM, J, M, N
parameter(IDIM = 3, JDIM = 5, M = 3, N = 2)
real D(IDIM, JDIM), UPARAM, ZERO
parameter(ZERO = 0.0E0)
data (D(1,J), J=1,5) / 0.870E0, 0.796E0, 1.0E0, 0.0E0, 0.0E0 /
data (D(2,J), J=1,5) / 0.571E0, -0.804E0, 0.0E0, 1.0E0, 0.0E0 /
data (D(3,J), J=1,5) / -0.960E0, 0.346E0, 0.0E0, 0.0E0, 1.0E0 /

print*, 'DRSHTCC... Demonstrate computation of a QR decomposition.'
print*, 'Given matrix A, compute orthogonal Q and triangular R'
print*, 'such that A = Q * R.'
call SMATP(D, IDIM, M, N, '0 A =')
do 20002 J=1,N
call SHTCC(1, J, J+1, M, D(1, J), UPARAM, D(1, J+1), IDIM, M+N-J)
20002 continue
call SMATP(D, 1, N+1, IDIM, M, '0 Q [ Transposed ] =')
D(2,1) = ZERO
D(3,1) = ZERO
D(3,2) = ZERO
call SMATP(D, IDIM, M, N, '0 R =')
stop
end

ODSHTCC

DRSHTCC... Demonstrate computation of a QR decomposition.
Given matrix A, compute orthogonal Q and triangular R
such that A = Q * R.

A =

<table>
<thead>
<tr>
<th>COL 1</th>
<th>COL 2</th>
</tr>
</thead>
<tbody>
<tr>
<td>ROW 1</td>
<td>0.8700000</td>
</tr>
<tr>
<td>ROW 2</td>
<td>0.5710000</td>
</tr>
<tr>
<td>ROW 3</td>
<td>-0.9600000</td>
</tr>
</tbody>
</table>

Q [ Transposed ] =

<table>
<thead>
<tr>
<th>COL 1</th>
<th>COL 2</th>
<th>COL 3</th>
</tr>
</thead>
<tbody>
<tr>
<td>ROW 1</td>
<td>-0.6144857</td>
<td>-0.4033004</td>
</tr>
<tr>
<td>ROW 2</td>
<td>0.7102542</td>
<td>-0.6569378</td>
</tr>
<tr>
<td>ROW 3</td>
<td>0.3434333</td>
<td>0.6370100</td>
</tr>
</tbody>
</table>

R =

<table>
<thead>
<tr>
<th>COL 1</th>
<th>COL 2</th>
</tr>
</thead>
<tbody>
<tr>
<td>ROW 1</td>
<td>-1.415818</td>
</tr>
<tr>
<td>ROW 2</td>
<td>0.0000000</td>
</tr>
<tr>
<td>ROW 3</td>
<td>0.0000000</td>
</tr>
</tbody>
</table>

6.4-4 One Householder Transformation July 11, 2015